

SERVO MOTOR YAPALIM


Servo motorlar dönme açısını kontrol edebildiğimiz motorlardır. "Servo" nun Türkçe karşılığı "köle" dir. Servo motorlar ile bir robotun kolunun veya bacağına açısını kontrol edebiliriz. Yazımızda ilk olarak hazır bulabileceğimiz hobi servo motorları nasıl kontrol edebileceğimizi inceleyelim, sonra da DC dişli kutulu bir motoru nasıl servo motor yapabiliriz öğreneelim.

Hobi servo motorlar (resim:1) genellikle 0-180 derece arasında dönebilirler. Çıkış torku 2 ile 25 kg.cm arasında servo motorlar bulmak mümkün. Model uçakların kanatlarını kontrol etmek gibi birçok iş için kullanılabilirler.



Resim: 1

3 adet girişten kırmızı olan +5 volt, siyah olan 0 volt (şasi) sarı olan da bilgi girişidir. Bilgi girişinden 1500 mikro saniye uzunluğunda bir sinyal gönderilirse motor 90 derece konumuna gider (resim:2). Motor, bu sinyalin genişliği 500 mikrosaniye ye yaklaştıkça 0 dereceye doğru, 2500 mikrosaniye ye yaklaştıkça da 180 dereceye doğru döner. Bu değerler değişiklik gösterebilir, deneyerek belirlenmelidir. Motorun belirtilen konumda kararlı bir şekilde kalabilmesi için bu sinyal sık sık uygulanmalıdır, 5 ila 100 milisaniye de bir uygulanabilir.




Resim: 2


Şimdi bu sinyali bir mikroişlemciye üretirip motoru 90 derece konumuna götürüelim. Devrede (resim:3) pic16f64 işlemcisi portb.1 bacağından bu sinyali üretmektedir. 7805 entegresi 3. bacağından 5 volt üretir. İşlemciyi programlamak için PicBasicPro dilinde yazılmış aşağıdaki kod kullanılabilir.

```
ANA: HIGH PORTB.1 ;portb.1 5V oldu
PAUSEUS 1500 ;1500 mikrosaniye bekle
;bu değer değiştirilerek
;açı kontrolü yapılır
LOW PORTB.1 ;portb.1 0V oldu
PAUSE 5 ;5 ms bekle
GOTO ANA
```

Genellikle bir şeyi hazır almaktan yapmaktan daha heyecan vericidir. Şimdi kendi servo motorumuzu yapalım. İlk olarak ihtiyacımızı karşılayacak tork ve hız değerinde DC dişli kutulu motor bulmalıyız. Böyle bir yapı hazır alınabildiği gibi oyuncaklardan da çıkarılabilir. İsterseniz bir doğru adım motorunu çarklarla yavaşlatarak kendi dişli kutulu motorunuzu yapabilirsiniz. Resim 4 te denemelerimde kullandığım DC dişli kutulu motor var. Açık kontrolü için dişli kutusunun ana çıkışından geri besleme almak gereklidir. Bunun için potansiyometre yada



Resim: 3



infrared okuyucular (shaft encoder) kullanılabilir. Biz denemelerimizde potansiyometre kullanacağız. Ana çıkış döndükçe potansiyometre de ona bağlı dönecek ve potansiyometrenin direnci değişecektir.

Tasarlayacağımız elektronik devre ve program ilk önce gidilmesi gereken açı bilgisini (hedef açı) başka bir potansiyometreden okuyacak, sonra dişli kutusu ana çıkışına bağlı potansiyometreyi okuyacak (açı). Eğer hedef açı, 0 anki açığa eşitse motorun çalıştırılmasına gerek yok. Eğer açı hedef açıdan büyükse motor hatayı kapatacak şekilde bir tarafa, küçükse diğer tarafa dönerek hedef açığa gidecek. Pratikte bu algoritmanın çalışmasında bazı problemler çıkacaktır. Motorun, dişlilerin ve ana çıkışa bağlı yükün ağırlığı motorun bir anda harekete geçmesini ve durmasını engeller. Açı, hedef açıdan büyük olduğunda Motor bir tarafa doğru dönerken açı, hedef açığa eşit olduğunda motora dur emri verirken motor sahip olduğu enerjiden dolayı


duramayacak ve açı hedef açıdan daha küçük olacaktır. Böylece hiç bitmeyen bir salınım (titreşim) başlayabilir. Bu durumda daha gelişmiş bir kontrol algoritması kullanılmalıdır. Mesela başlangıç için P kontrol (proportional control - oran kontrolü) kullanılabilir. Bu yöntemde açı ile hedef açı arasındaki fark (hata) ne kadar büyükse o kadar büyük motor voltajı uygulanır.

$$V_{\text{motor}} = \text{sabit} \cdot \text{hata}$$

Böylece açı hedef açığa yaklaştıkça motor yavaşladığından sistemin salınım girmesi kısmen engellenmiş olur. P kontrolün dezavantajı küçük hataları giderememesidir. Açı hatası çok az olduğunda motora çok az bir voltaj uygulanacak ve motor sürtünmelerden dolayı dönmeyecektir. Bu küçük hatayı gidermek için I kontrol (integral kontrolü) eklenebilir ve PI kontrol elde edilir.

$$V_{\text{motor}} = \text{sabit1} \cdot \text{hata} + \text{sabit2} \cdot \int \text{hata}$$

Eğer motorun dönmeyeceği kadar küçük bir hata olursa, bu hatanın zamana göre integrali alındığından motor voltajı artacak ve motor hatayı kapatacak bir dönme yapacaktır. Sabitle doğru seçmek çok önemlidir, örneğin sabit2 (integral sabiti) büyük seçilirse sistem yine salınım girer. PID kontrol daha etkili bir yöntemdir. Yazımızda sadece P ve PI kontrol deneyeceğiz. Şimdi motoru kontrol edeceğimiz devreyi inceleyelim (resim:5).



Resim: 5

```
PAY CON 3 ;hata payı, 1=0.25 derece
;salınımları azaltmak için hata payı artırılabilir
DEFINE ADC_BITS 10 ;10 bit analog-dijital çevirim
DEFINE ADC_CLOCK 2
ADCON0=%10000001 ;analog-dijital çevirim ayarları
ADCON1=%10000010 ;//
TRISA=255 ;porta giriş yapıldı
TRISC=0 ;portc çıkış yapıldı
PR2=99 ;2,5 Kilo hertz PWM e karşılık geliyor
T2CON=%00000101 ;//
CCP1CON=%1100 ;PWM modu açık
CCP2CON=%1100 ;//
T1CON=%00000001 ;timer1 çalışıyor
ACI VAR WORD ;o anki açı bilgisi burda saklanır
HEDEFACI VAR WORD ;hedef açı bilgisi
TEPKI VAR BYTE ;motora verilecek tepki (voltaj)
FARK VAR BYTE ;hata
INTEGRAL VAR BYTE ;integral sabiti
;Not: Eğer CCP1L değişkenine 0 yazılırsa motor durur, 99 yazılırsa motor tam
;hızla çalışır. Aradaki değerler ara hızlara karşılık gelir. CCP2L değişkenine
; yazılırsa motor geri döner. Sorular için: deniz@robot.metu.edu.tr
ANA:
ADCIN 0,ACI ;açı potansiyometresini oku (motora bağlı)
ADCIN 1,HEDEFACI ;hedef açı potansiyometresini oku
IF ACI>HEDEFACI+PAY THEN
FARK=ACI-HEDEFACI
IF ACI>HEDEFACI+99 THEN FARK=99
TEPKI=FARK+INTEGRAL ;P ve I ekleme
IF TEPKI>99 THEN TEPKI=99 ;99 motorun en yüksek hızına karşılık
IF TEPKI<30 THEN TEPKI=30 ;30 motorun en düşük hızına karşılık
CCPR1L=TEPKI ;motor bir tarafa doğru (teпки/99)
CCPR2L=0 ;hızıyla döner
ENDIF
IF HEDEFACI>ACI+PAY THEN
FARK=HEDEFACI-ACI
IF HEDEFACI>ACI+99 THEN FARK=99
TEPKI=FARK-INTEGRAL
IF TEPKI>99 THEN TEPKI=99
IF TEPKI<30 THEN TEPKI=30
CCPR1L=0
CCPR2L=TEPKI ;motor diğer tarafa döner
ENDIF
IF PIR1.0=1 THEN ;10 mili saniyede bir buraya girer
TMR1H=216 ;eger açı artırılırsa timer1 daha çabuk taşar
INTEGRAL=INTEGRAL+1 ;integral değeri daha sık artırılır,
IF INTEGRAL>99 THEN INTEGRAL=99
PIR1.0=0 ;timer1 bayrağı kuruldu
ENDIF
IF ACI<HEDEFACI+PAY AND ACI>=HEDEFACI-PAY THEN
CCPR1L=0 ;eger açı uygun aralıktaysa integral sıfırlanır
CCPR2L=0 ;ve motor durdurulur
INTEGRAL=0
ENDIF
GOTO ANA ;sonsuz döngü
END
```

Potansiyometre okuyabilmek için analog sinyaller dijitalle çevrilmeli. Bu devrede analog-dijital çeviricisi olan pic16f877 işlemcisi kullanıldı. Dişli kutulu motor çıkışına bağlı potansiyometrenin ve hedef açı belirleyen potansiyometrenin çıkışı işlemcinin analog kanallarına bağlandı. Motoru sürmek için de 2 ampere kadar olan yükleri sürebilen L298 entegresi kullanıldı. Aşağıda PI kontrol ile motorun açısını kontrol eden kod görülebilir. Motorun hızı PWM (pulse width modulation - sinyal genişlikli modülasyon) tekniğiyle kontrol edilmektedir.

ODTÜ Robot Topluluğu
<http://robot.metu.edu.tr>
 Mustafa Deniz
www.mustafadeniz.com
deniz@robot.metu.edu.tr